
$TITLE: M10-4a.gms
*CALIBRATES MODEL TO SHEET IO-1 IN M10-IOTABLE.XLS
*TRADE SURPLUS OF 5
*NETS OUT CROSS-HAULING: domestic and foreign goods perfect substitutes
*acutal MAN exports 20, imports 10; AGR exports 5, imports 10

$ONTEXT
              Ag     Man    Ser    Cons  Exports  Imports
               1     2      3      4     5        6    UM(1:5) - (6)

Ag     1       5     6      9     30              5  | 45
Man    2      10    15     20     60    10           |115
Ser    3      12    13      5    110                 |140
Wages  4       9    30     66                        |105
Rents  5       9    51     40                        |100
              _______________________________________
              45   115    140    200    10       -5

Micro consistent: total value added (205 = 105+100) =
consumption + trade surplus (205 = 200 + 10 - 5)

        Net Ag   Net Man  Net Ser  Export  Imports   Welf   Cons  Row
Ag        40       -6       -9                5      -30         | 0
Man      -10      100      -20        -10            -60         | 0
Ser      -12      -13      135                      -110         | 0
Wages     -9      -30      -66                             105   | 0
Rents     -9      -51      -40                             100   | 0
For ex                                 10    -5             -5   | 0
Welfare                                              200  -200   | 0
          _______________________________________________________
Col sums   0        0        0         0      0        0     0

$OFFTEXT

PARAMETERS
 P W A  world price of agriculture
 P W M  world price of manufactures;

PWA = 1;
PWM = 1;

$ONTEXT

$MODEL:M10_4

$SECTORS:

 AGR
 MAN
 SER
 IMAGR EXAGR
 EXMAN IMMAN
 WELFARE

$COMMODITIES:
 PAGR
 PMAN
 PSER
 PL
 PK
 PFX
 PWEL

$CONSUMERS:
 CONS

$PROD:AGR   s:1
 O:PAGR  Q:45
 I:PAGR  Q:5
 I:PMAN  Q:10
 I:PSER  Q:12
 I:PL    Q:9
 I:PK    Q:9

$PROD:MAN   s:1
 O:PMAN  Q:115
 I:PAGR  Q:6
 I:PMAN  Q:15
 I:PSER  Q:13
 I:PL    Q:30
 I:PK    Q:51

$PROD:SER   s:1
 O:PSER  Q:140
 I:PAGR  Q:9
 I:PMAN  Q:20
 I:PSER  Q:5
 I:PL    Q:66
 I:PK    Q:40

$PROD:IMAGR
 O:PAGR  Q:5
 I:PFX   Q:(5*PWA)

$PROD:EXMAN
 O:PFX   Q:(10*PWM)
 I:PMAN  Q:10

$PROD:EXAGR

 O:PFX   Q:(5*PWA)
 I:PAGR  Q:(5*1.001)

$PROD:IMMAN
 O:PMAN  Q:(10*0.999)
 I:PFX   Q:(10*PWM)

$PROD:WELFARE  s:1
 O:PWEL  Q:200
 I:PAGR  Q:30
 I:PMAN  Q:60
 I:PSER  Q:110

$DEMAND:CONS
 D:PWEL
 E:PL    Q:105
 E:PK    Q:100
 E:PFX   Q:(-5)

$OFFTEXT

$SYSINCLUDE MPSGESET M10_4

PWEL.FX = 1;
EXAGR.L = 0;
IMMAN.L = 0;

M10_4.ITERLIM = 0;
$INCLUDE M10_4.GEN
SOLVE M10_4 USING MCP;

M10_4.ITERLIM = 2000;
$INCLUDE M10_4.GEN
SOLVE M10_4 USING MCP;

* counterfactual: double the world price of the export good M

PWM = 2;

$INCLUDE M10_4.GEN
SOLVE M10_4 USING MCP;

* counterfactual: double the world price of the import good A

PWM = 1;
PWA = 2;

$INCLUDE M10_4.GEN
SOLVE M10_4 USING MCP;


